Wstęp do algebry
Zajęcia nr 4

Sprzeżenie hermitowskie macierzy

Sprzeżeniem hermitowskim macierzy A nazywamy macierz $A^{\dagger}$

$A^{\dagger}=\overline{A}^{T}$

$(A^{\dagger})_{ij}=\overline{A}_{ji}$

jezeli A==$A^{\dagger}$ to macierz A jest hermitowska (samosprzężona)

```
A = {{1, -I, 7}, {7, 5, 8 I}, {5 I + 1, 6, 5}}
```

{{1, -ⅈ, 7}, {7, 5, 8 ⅈ}, {1 + 5 ⅈ, 6, 5}}

```
A // MatrixForm
```

$$\begin{pmatrix} 1 & -i & 7 \\ 7 & 5 & 8i \\ 1+5i & 6 & 5 \end{pmatrix}$$

Macierz powstała przez sprzeżenie zespolone elementow macierzy A:

```
Conjugate[A] // MatrixForm
```

$$\begin{pmatrix} 1 & i & 7 \\ 7 & 5 & -8i \\ 1-5i & 6 & 5 \end{pmatrix}$$

Macierz powstała przez sprzeżenie zespolone elementow macierzy A a nastepnie jej transponowanie:

```
Transpose[Conjugate[A] ] // MatrixForm
```

$$\begin{pmatrix} 1 & 7 & 1-5i \\ i & 5 & 6 \\ 7 & -8i & 5 \end{pmatrix}$$

Sprzężenie hermitowskie macierzy A mozemy uzyskac przy pomocy jednej funkcji ConjugateTranspose[]

```
ConjugateTranspose[A] // MatrixForm
```

$$\begin{pmatrix} 1 & 7 & 1-5\,i \\ i & 5 & 6 \\ 7 & -8\,i & 5 \end{pmatrix}$$

Czy macierz A jest hermitowska (samosprzeżona)?

```
ConjugateTranspose[A] == A
```

```
False
```

```
HermitianMatrixQ[A]
```

```
False
```

Iloczyn skalarny

Zdefiniujmy iloczyn skalarny w zespolonej przestrzni liniowej:

```
Scalar[a_, b_] = a.Conjugate[b]
```

```
a.Conjugate[b]
```

jest on linowy w pierwszym argumencie i antyliniowy w drugim

$(\alpha\, a\,,\, b\,) = \alpha\,(\,a\,,\,b\,)$

$(\,a\,,\alpha\, b\,) = \overline{\alpha}(\,a\,,\,b\,)$

Zadanie 1

Zdefiniuj macierz

B= $\begin{pmatrix} 1 & -i & 1 \\ 2 & 1+2\,i & i \\ 1 & 1+i & 1 \end{pmatrix}$

Znajdź macierz sprzeżoną $B^{\dagger}$.

Czy macierz B jest hermitowska (samosprzeżona)?

Zdefiniuj 2 przykładowe wektory 3-wymiarowe, np w={1,0,i} oraz v={0,2 i, 1}.

Sprawdź że zachodzi:

(B v, w)=(v, $B^{\dagger}$w)

Zadanie2

Zdefiniuj macierz

M= $\begin{pmatrix} a1 + i\, a2 & b1 + i\, b2 \\ c1 + i\, c2 & d1 + i\, d2 \end{pmatrix}$

liczby a_i, b_i, c_i, d_i sa rzeczywiste. Jest to ogólna postac macierzy o elementach zespolonych.
Znajdź $M^\dagger$ (ComplexExpand[] i ConjugateTranspose[])
Sprawdzic warunek, dla ktorego macierz jest hermitowska, czyli M==$M^\dagger$ (funkcja Solve[])

MacierzePauliego

Można pokazać, że taka macierz jest kombinacją liniową macierzy jednostkowej $\begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix}$ i macierzy Pauliego $\begin{pmatrix} 0 & 1 \\ 1 & 0 \end{pmatrix}$, $\begin{pmatrix} 0 & -i \\ i & 0 \end{pmatrix}$, $\begin{pmatrix} 1 & 0 \\ 0 & -1 \end{pmatrix}$

```
Simplify[(a1 + d1) / 2 IdentityMatrix[2]  + b1 PauliMatrix[1]  -
   b2 PauliMatrix[2]  + (a1 - d1) / 2 PauliMatrix[3] ]  // MatrixForm
```

Każdą macierz hermitowską 2x2 można wyrazić jako kombinację linową macierzy jednostkowej i macierzy Pauliego, gdzie współczynniki sa liczbami rzeczywistymi

```
PauliMatrix[1] // MatrixForm
```

$$\begin{pmatrix} 0 & 1 \\ 1 & 0 \end{pmatrix}$$

```
PauliMatrix[2] // MatrixForm
```

$$\begin{pmatrix} 0 & -i \\ i & 0 \end{pmatrix}$$

```
PauliMatrix[3] // MatrixForm
```

$$\begin{pmatrix} 1 & 0 \\ 0 & -1 \end{pmatrix}$$

Zadanie 3

Pokazać, że macierze Pauliego mają wlasnosci:
det σ_k =-1 (Det[])
Tr σ_k =0 (Tr[])

$\sigma_k = \sigma_k{}^\dagger$ (hermitowskość) (ConjugateTranspose[])
$\sigma_k{}^\dagger = \sigma_k{}^{-1}$ (unitarność) (Inverse[])

Oblicz iloczyn dwóch takich samych macierzy Pauliego:
$\sigma_k \sigma_k$ (Mnożenie macierzy to A.B lub Dot[A,B], n-ta potęga macierzy MatrixPower[A,n])

Pytanie: Czemu bedzie sie równać wyrażenie $(\sigma_k)^N$ gdy N jest liczbą parzystą/liczbą nieparzystą?

```
? MatrixPower
```

MatrixPower[m, n] gives the n^{th} matrix power of the matrix m.
MatrixPower[m, n, v] gives the n^{th} matrix power of the matrix m applied to the vector v. >>

```
? Dot
```

$a.b.c$ or Dot[a, b, c] gives products of vectors, matrices, and tensors. >>

Zadanie 4

Sprawdzić własność:

$$-i\,\sigma_1\sigma_2\sigma_3 = \begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix}$$

Zadanie 5

Sprawdzic własność:

$$\sigma_x\sigma_y = i\sum_{z=1}^{3}\epsilon_{xyz}\sigma_z + \delta_{xy}\begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix}$$

dla:
a) x=1, y=2
b) x=1, y=3
c) x=1, y=1

δ_{xy} to delta Kroneckera   KroneckerDelta[x,y]

$$\delta_{xy}=\begin{cases} 1 & \text{gdy } x = y \\ 0 & \text{gdy } x \neq y \end{cases}$$

ϵ_{xyz} to symbol Leviego-Civity   Signature[{x,y,z}]

$$\epsilon_{xyz} = \begin{cases} 1 & \text{gdy } x\ y\ z \text{ to permutacja parzysta (np 123)} \\ 0 & \text{gdy } x = y,\ y = z \text{ lub } z = x \\ -1 & \text{gdy } x\ y\ x \text{ to permutacja nieparzysta (np 213)} \end{cases}$$

```
Signature[{1, 2, 3}]
Signature[{2, 1, 3}]
Signature[{1, 1, 3}]
```

1

-1

0

```
KroneckerDelta[1, 1]
KroneckerDelta[1, 2]
```

1

0

```
? KroneckerDelta
```

KroneckerDelta[n_1, n_2, …] gives the Kronecker delta $\delta_{n_1 n_2 \ldots}$, equal to 1 if all the n_i are equal, and 0 otherwise. >>

```
? DiracDelta
```

DiracDelta[x] represents the Dirac delta function $\delta(x)$.
DiracDelta[x_1, x_2, …] represents the multidimensional Dirac delta function $\delta(x_1, x_2, \ldots)$. >>

```
? Sum
```

Sum[f, {i, i_{max}}] evaluates the sum $\sum_{i=1}^{i_{max}} f$.
Sum[f, {i, i_{min}, i_{max}}] starts with $i = i_{min}$.
Sum[f, {i, i_{min}, i_{max}, di}] uses steps di.
Sum[f, {i, {i_1, i_2, …}}] uses successive values i_1, i_2, … .
Sum[f, {i, i_{min}, i_{max}}, {j, j_{min}, j_{max}}, …] evaluates the multiple sum $\sum_{i=i_{min}}^{i_{max}} \sum_{j=j_{min}}^{j_{max}} \ldots f$.
Sum[f, i] gives the indefinite sum $\sum_i f$. >>

Zadanie 6

Wprowadźmy pojęcie komutatora:
[a,b]=a b - b a
oraz antykomutatora
{a,b}=a b+b a

Zdefiniuj funkcje dwóch zmiennych Komutator[a,b] oraz AntyKomutator[a,b] , które pobierają macierze i zwracają ich komutator/antykomutator.

Sprawdzic własność :

Komutator[σ_x, σ_y]=2 i $\sum_{z=1}^{3} \epsilon_{xyz} \; \sigma_z$
dla
a) x=1, y=2
b) x=1, y=3
c) x=1, y=1

AntyKomutator[σ_x, σ_y]=2 $\delta_{xy}\begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix}$
dla
a) x=1, y=2
b) x=1, y=3
c) x=1, y=1

Macierze Pauliego a SU(2)

```
u1[x_] := I x PauliMatrix[1]
u2[x_] := I x PauliMatrix[2]
u3[x_] := I x PauliMatrix[3]
```

```
u1[x] // MatrixForm
u2[x] // MatrixForm
u3[x] // MatrixForm
```

$$\begin{pmatrix} 0 & i\,x \\ i\,x & 0 \end{pmatrix}$$

$$\begin{pmatrix} 0 & x \\ -x & 0 \end{pmatrix}$$

$$\begin{pmatrix} i\,x & 0 \\ 0 & -i\,x \end{pmatrix}$$

```
? MatrixExp
```

MatrixExp[*m*] gives the matrix exponential of *m*.
MatrixExp[*m*, *v*] gives the matrix exponential of *m* applied to the vector *v*. >>

Zdefinujmy funkcje postaci $e^{i\,x\,\sigma_k}$

```
e1[x_] := MatrixExp[u1[x]]
e2[x_] := MatrixExp[u2[x]]
e3[x_] := MatrixExp[u3[x]]
```

```
e1[x] // MatrixForm
e2[x] // MatrixForm
e3[x] // MatrixForm
```

$$\begin{pmatrix} \text{Cos}[x] & i\,\text{Sin}[x] \\ i\,\text{Sin}[x] & \text{Cos}[x] \end{pmatrix}$$

$$\begin{pmatrix} \text{Cos}[x] & \text{Sin}[x] \\ -\text{Sin}[x] & \text{Cos}[x] \end{pmatrix}$$

$$\begin{pmatrix} e^{i\,x} & 0 \\ 0 & e^{-i\,x} \end{pmatrix}$$

Zadanie 7

Pokaż, że macierze e1[x], e2[x], e3[x] :
sa unitarne tzn: $A^{-1} = A^{\dagger}$
i ich wyznacznik wynosi 1